Maple-assisted proof of formula for A181209

Robert Israel

25 December 2017

Of the $2^{10} = 1024$ possible configurations for a 2×5 sub-array, 169 have no 1's diagonally or antidiagonally adjacent. Consider the 169×169 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×5 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j), and 0 otherwise. The following Maple code computes it.

```
> Configs:= select(A -> A[1]*A[7]=0 and A[2]*A[8] = 0 and A[3]*A[9]
  =0 and A[4]*A[10]=0 and A[2]*A[6]=0 and A[3]*A[7]=0 and A[4]*A[8]
  =0 and A[5]*A[9]=0,
  [seq(convert(x,base,2)[1..10],x=2^10..2^11-1)]): nops(Configs);
```

$$169 \qquad (1)$$

```
> T:= Matrix(169,169,(i,j) -> `if`(Configs[i][6..10] = Configs[j]
  [1..5],1,0));
```

$$T := \left[\begin{array}{c} \textit{169 x 169 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{array}\right] \qquad (2)$$

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](169, i -> `if`(Configs[i][1..5] = [0,0,0,0,0],1,
  0)):
  v:= Vector(169, j -> `if`(Configs[j][6..10] = [0,0,0,0,0],1,0)):
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v,n=1..10);
```

$$32, 169, 2117, 17424, 177073, 1630729, 15786848, 149352841, 1429585373, 13610488896 \qquad (3)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{18} - 13\, t^{17} + 439\, t^{15} - 919\, t^{14} - 3792\, t^{13} + 11886\, t^{12} + 6450\, t^{11} - 44016\, t^{10} + 14274\, t^9 + 59430\, t^8 - 42240\, t^7 - 22975\, t^6 + 24475\, t^5 - 3625\, t^3 + 625\, t^2 \qquad (4)$$

This turns out to have degree 18. Thus we will have $0 = u\, P(T)\, T^n\, v = \sum_{i=0}^{18} p_i\, a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 18, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 9, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^2\, (t-1)\, (t^3 - 12\, t^2 + 24\, t - 5)\, (t^6 - 12\, t^4 + 24\, t^2 - 5)\, (t^6 - 24\, t^4 + 60\, t^2 - 25) \qquad (5)$$

```
> Q:= unapply(t^9-12*t^8+283*t^6-516*t^5-600*t^4+1415*t^3-600*
  t+125, t);
```

$$Q := t \to t^9 - 12\,t^8 + 283\,t^6 - 516\,t^5 - 600\,t^4 + 1415\,t^3 - 600\,t + 125 \tag{6}$$

The complementary factor $R(t) = \frac{P(t)}{Q(t)}$ has degree 9.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to \left(t^7 - t^6 - 12\,t^5 + 12\,t^4 + 24\,t^3 - 24\,t^2 - 5\,t + 5\right)\,t^2 \tag{7}$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-9 recurrence

$$\sum_{i=0}^{9} r_i\,b(i+n) = \sum_{i=0}^{9} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(8) = 0$. This might take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 8 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..8);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0 \tag{8}$$